
ME 410    Day 27 
 

• Description of Finite Heat Release Function - SI Engines 
• Differential Equations to Model Cycle 
• Software Implementation in EES 
• Questions that can be answered 

 
1. Description of Finite Heat Release Function 

 
This is an attempt to model more realistically the burn rates of fuel in 
the cylinder during the working cycle.  The key is to specify heat 
addition rate as a function of crank angle. 
 
Naturally, heat addition rate is zero for much of the cycle.  We start 
with an “S” curve or Weibe function which describes the fraction of 
fuel that has been burned. 
 

 
Here is the math function which describes the burned fraction. 

The shape is determined by the parameters a and n.  Values of a = 5 
and n = 3 have been found to agree well with experiments. See 
pages 766 to 769 in the Text.  Another ref:  Internal Combustion 
Engines by Ferguson and Kirkpatrick, Wiley 2001. 

How it can be used. 

This is a very useful description of how heat input rate varies with θ 
during combustion.  Naturally, when the combustion process is not 
occurring, this derivative is zero. 
 
 

2. Differential Equations to Model Cycle 
 
Start with the Ideal Gas Law 
 
  mRTPV =  
 
and differentiate with respect to q. 

Next, write the first law of thermo for this closed system in rate form. 
 
  dUWQ =δ−δ  
 
and make the following substitutions. 
 
  dVPW =δ  
 
  dTcmdU v=  
 



After some hand waving, 

Next solve for the derivative of pressure with respect to θ. 

Next, we need the following expressions from kinematics.  See our 
text, page 44. 

Now R no longer stands for the gas constant, it now stands for the 
ratio of connecting rod length to crank radius.  Vc is the clearance 
volume and V the volume of the cylinder at θ. 
 
Since ( ) ccd V1rV −=  we can write this as 

The rate at which volume is swept out as a function of crank angle, 
we get by differentiating with respect to θ. 

Finally, we recall the definition of work gives us 

3. Software Implementations 

 
We conceptualize this problem mathematically as 4 simultaneous 
differential equations in the independent variable q and having 
dependent variables V, P, T and W. 
 
These equations must then be solved numerically using a time 
stepping scheme. 
 
The “integral” function in EES invokes the time stepping scheme 
which is described in the software help. 
 
 
The EES file is included in these notes. 
 
 
 
 
 
 



"Finite Heat Release Calculations" 
 
"Reference:  Internal Combustion Engines by Colin Ferguson and 
Allan Kirkpatrick,  2nd Edition. John Wiley 2001.  Pages 39 44." 
 
"The heat release occurs over a finite time given by a Weibe 
function.  The  heat release rate is calculated and used to write 
the first law of thermo in rate form. (Rate of change with respect to 
crank angle)  The ideal gas law is used to relate pressure and 
temperature rates.  The kinematics of the engine give volumetric 
rates.  The result is a set of differential equations which may be 
integrated to describe the cycle." 
 
"Function to calculate burn fraction vs. crank angle. Crank angle 
varies from -pi at start of cycle to 0 at TC to pi at the end of the 
cycle.  Exhaust processes are not modeled." 
 
function xb(theta) 
{ theta_s is start of combustion in radians, theta_d is the duration, 
a and n are the  
  Weibe parameters.}     
$common  theta_s, theta_d,a,n 
      if (theta < theta_s)  then 
              xb := 0 
     else 
              if (theta > theta_s + theta_d) then 
                       xb := 1 
             else 
                       xb := 1-exp(- a*((theta-theta_s)/theta_d)^n) 
             endif 
      endif 
end 
 
"Function to calculate heat release rate" 
 
function dQdtheta(theta) 
{ theta_s is start of combustion in radians, theta_d is the duration, 
a and n are the  
  Weibe parameters. Q is the total heat release } 
$common  theta_s, theta_d,a,n,Q 



      if (theta < theta_s) then 
              dQdtheta := 0 
      else 
              if (theta > theta_s+theta_d) then 
                      dQdtheta := 0 
              else 
                      dQdtheta := n*a*Q/theta_d*(1-xb(theta))*((theta-
theta_s)/theta_d)^(n-1) 
             endif 
     endif 
end 
     
"Function to calculate volume rate wrt crank angle" 
 
function dVdtheta(theta)  
$common V_d, r_c, R 
dVdtheta := V_d/2 * sin(theta)*(1+cos(theta)/sqrt(R^2 -
sin(theta)^2)) 
end 
 
"Function to calculate volume at a given crank angle" 
 
function V(theta) 
$common V_d, r_c, R 
V := V_d/(r_c-1) + V_d/2 *(R + 1 - cos(theta) - sqrt(R^2-
sin(theta)^2)) 
end 
 
"Combustion Data" 
 
theta_s = -pi/9 
theta_d = 4*pi/9 
a = 5 
n = 3 
Q = 1.8 
 
"Engine Data" 
 
bore = 0.1 
stroke = 0.1 



P_1 = 101.3 
r_c = 10 
V_d = pi /4 * bore^2 * stroke 
L = 0.150 
R = 2*L/stroke 
gamma = 1.4 
T_1 = 300 
V_1 = V(theta_1) 
mR = P_1*V_1 / T_1 
Mw = 29 
theta_1 = -pi 
theta_f = pi 
 
vv = V(theta) 
 
"The derivatives of Pressure and Temperature, Work and Vol with 
respect to crank angle" 
 
dPdtheta = -gamma*P/V(theta) * dVdtheta(theta) + (gamma-
1)/V(theta)*dQdtheta(theta) 
dTdtheta = 1/mR * ( P * dVdtheta(theta) + V(theta)*dPdtheta ) 
dWdtheta = P * dVdtheta(theta) 
dVoldtheta = dVdtheta(theta) 
 
"We can now integrate the above differential equations." 
 
P = P_1 + integral(dPdtheta,theta,theta_1,theta_f) 
T = T_1 + integral(dTdtheta,theta,theta_1,theta_f) 
W =           integral(dWdtheta,theta,theta_1,theta_f) 
Vol = V_1 + integral(dVoldtheta,theta,theta_1,theta_f) 
 
$IntegralTable theta:0.1,P,T,W,Vol 
 
"Performance Calculations" 
 
eta_f = W / Q 
imep = W / V_d 
 
 
 



 
4. Questions that can be answered 

 
This model is useful in the sense that it allows us to study the effects 
of spark timing on engine performance. 
 
 


